```
" DYNAMIC FORCE ANALYSIS VIA DYADS "
Apply [Clear, Names["Global`*"] ] ;

Off[General::spell];
Off[General::spell1];

(* Input data *)
n = 50 ; (* rpm *)
omega = n N[Pi]/30 ; (* rad/s *)

rule = {AB->0.14, AC->0.06, AE->0.25, CD->0.15, FD->0.4, EG->0.5, h->0.01, d->0.001,
hSlider->0.02, wSlider->0.05, rho->8000, g->9.807, Me->-100., phi[t]->N[Pi]/6, phi'[t]-
>omega, phi''[t]->0} ;

(* Position analysis *)

(* Position of joint A *)
xA = yA = 0;
rA = { xA, yA, 0};
Print["rA = ", rA, " [m]"] ;

(* Position of joint C *)
xC = 0 ;
yC = AC ;
rC = { xC, yC, 0} ;
Print["rC = AC = ", rC/.rule, " [m]"] ;

(* Position of joint E *)
xE = 0 ;
yE = -AE ;
rE = { xE, yE, 0};
Print["rE = AE = ", rE/.rule, " [m]"] ;


(* Position, velocity and acceleration of joint B *)
xB = AB Cos [ phi[t] ] ;
yB = AB Sin [ phi[t] ] ;
rB = { xB, yB, 0} ;
Print["rB = AB = ", rB/.rule, " [m]"] ;
vB = D[rB,t] ;
aB = D[D[rB,t],t] ;

(* Position, velocity and acceleration of joint D *)

(* Parameters m and n of line BC: y = m x + n *)
mBC = ( yB - yC ) / ( xB - xC ) ;
nBC = yB - mBC xB ;
eqn41 = ( xDsol - xC )^2 + ( yDsol - yC )^2 - CD^2 == 0 ;
eqn42 = yDsol - mBC xDsol - nBC == 0 ;
solutionD = Solve [ { eqn41 , eqn42 } , { xDsol , yDsol } ];
(* Two solutions for D *)
xD1 = xDsol /. solutionD[[1]];
yD1 = yDsol /. solutionD[[1]];
xD2 = xDsol /. solutionD[[2]];
yD2 = yDsol /. solutionD[[2]];
(* Select the correct position for D *)
If[ (xD1/.rule)<=xC, xD=xD1; yD=yD1, xD=xD2; yD=yD2 ] ;
rD = { xD, yD, 0} ;
Print["rD = AD =", rD/.rule, " [m]"] ;
vD = D[rD,t] ;
aD = D[D[rD,t],t] ;

(* Angular velocity and acceleration of the link 1 *)
```

```
alpha1 = {0, 0, phi''[t]} ;

(* Angular velocity and acceleration of the link 2 and link 3 *)
phi2 = ArcTan[ mBC ] ;
alpha2 = {0, 0, D[D[phi2,t],t]} ;
phi3 = phi2 ;
alpha3 = alpha2 ;

(* Angular velocity and acceleration of the link 4 and link 5 *)

phi4 = ArcTan[(yD-yE)/(xD-xE)] + N[Pi] ;
alpha4 = {0, 0, D[D[phi4,t],t]} ;
phi5 = phi4 ;
alpha5 = alpha4 ;

(* -------------------------- *)
(* Inertia forces and moments *)
(* -------------------------- *)

(* Link 1 *)
m1 = rho AB h d /.rule ;
rC1 = rB/2 ;
Print["rC1 = AC1 =", rC1/.rule, " [m]"] ;
vC1 = vB/2 ;
aC1 = aB/2 ;
Fin1 = - m1 aC1 /.rule ;
G1 = {0, -m1*g, 0} /.rule ;
F1 = ( Fin1 + G1 ) /.rule ;
IC1 = m1 (AB^2+h^2)/12 /.rule ;
M1 = Min1 = - IC1 alpha1 /.rule ;
Print["F1 = ", F1, " [N]"] ;
Print["M1 = ", M1, " [Nm]"] ;

(* Link 2 *)
m2 = rho hSlider wSlider d /.rule ;
rC2 = rB ;
Print["rC2 = AC2 = AB = ", rC2/.rule, " [m]"] ;
vC2 = vB ;
aC2 = aB ;
Fin2 = - m2 aC2 /.rule ;
G2 = {0, -m2*g, 0} /.rule ;
F2 = ( Fin2 + G2 ) /.rule ;
IC2 = m2 (hSlider^2+wSlider^2)/12 /.rule ;
M2 = Min2 = - IC2 alpha2 /.rule ;
Print["F2 = ", F2, " [N]"] ;
Print["M2 = ", M2, " [Nm]"] ;

(* Link 3 *)
m3 = rho FD h d /.rule ;
xC3 = xC + (FD/2-CD) Cos [ phi3 ] ;
yC3 = yC + (FD/2-CD) Sin [ phi3 ] ;
rC3 = { xC3, yC3, 0 } ;
Print["rC3 = AC3 = ", rC3/.rule, " [m]"] ;
vC3 = D[rC3,t] ;
aC3 = D[D[rC3,t],t] ;
Fin3 = - m3 aC3 /.rule ;
G3 = {0, -m3*g, 0} /.rule ;
F3 = ( Fin3 + G3 ) /.rule ;
IC3 = m3 (FD^2+h^2)/12 /.rule ;
M3 = Min3 = - IC3 alpha3 /.rule ;
Print["F3 = ", F3, " [N]"] ;
Print["M3 = ", M3, " [Nm]"] ;

(* Link 4 *)
m4 = rho hSlider wSlider d /.rule ;
```

```
rC4 = rD ;
Print["rC4 = AC4 = AD =", rC4/.rule, " [m]"] ;
vC4 = vD ;
aC4 = aD ;
Fin4 = - m4 aC4 /.rule ;
G4 = {0, -m4*g, 0} /.rule ;
F4 = ( Fin4 + G4 ) /.rule ;
IC4 = m4 (hSlider^2+wSlider^2)/12 /.rule ;
M4 = Min4 = - IC4 alpha4 /.rule ;
Print["F4 = ", F4, " [N]"] ;
Print["M4 = ", M4, " [Nm]"] ;

(* Link 5 *)
m5 = rho EG h d /.rule ;
xC5 = EG/2 Cos [ phi5 ] ;
yC5 = EG/2 Sin [ phi5 ] ;
rC5 = { xC5, yC5, 0} ;
Print["rC5 = AC5 = ", rC5/.rule, " [m]"] ;
vC5 = D[rC5,t] ;
aC5 = D[D[rC5,t],t] ;
Fin5 = - m5 aC5 /.rule ;
G5 = {0, -m5*g, 0} /.rule ;
F5 = ( Fin5 + G5 ) /.rule ;
IC5 = m5 (EG^2+h^2)/12 /.rule ;
M5 = Min5 = - IC5 alpha5 /.rule ;
M5e = { 0, 0, Me } /.rule ;
Print["F5 = ", F5, " [N]"] ;
Print["M5 = ", M5, " [Nm]"] ;



(* --------------- *)
(* Joint reactions *)
(* --------------- *)

"Dyad 4 & 5"

" Joint E_R: F05 "
F05Sol = { F05xSol, F05ySol, 0 } ;
" Joint D_R: F34"
F34Sol = { F34xSol, F34ySol, 0 } ;

"∑F(4,5)=0: F05+F5+F4+F34=0 (1)"
eqF45 = F05Sol+F5+F4+F34Sol;
eqF45x = eqF45[[1]]==0;
eqF45y = eqF45[[2]]==0;

"∑F(5).DE=0: (F05+F5).DE=0 (2)"
rDE = ( rE - rD ) /.rule ;
eqF5DE = (F5+F05Sol).rDE == 0 ;

"∑M(4&5)_D=0: DExF05+DC5xF5+M4+M5+M5e=0 (3)"
rDC5 = ( rC5 - rD ) /.rule ;
eqM45D = (Cross[rDE,F05Sol]+Cross[rDC5,F5]+M4+M5+M5e)[[3]] == 0 ;

solF45 = Solve[{eqF45x,eqF45y,eqF5DE,eqM45D},
{F05xSol,F05ySol,F34xSol,F34ySol}];

F05x=F05xSol/.solF45[[1]];
F05y=F05ySol/.solF45[[1]];
F05={F05x,F05y,0};
F34x=F34xSol/.solF45[[1]];
F34y=F34ySol/.solF45[[1]];
F34={F34x,F34y,0};
F43=-F34;
```

```
Print["Eqs.(1)(2)(3) =>"];
Print["F05 = ",F05," [N]"];
Print["F34 = ",F34," [N]"];
"∑F(5)=0 => F45=-F05-F5 (4)"
F45=-F05-F5;
Print["F45 = ",F45," [N]"];

"Dyad 2 & 3"

" Joint C_R: F03 "
F03Sol = { F03xSol, F03ySol, 0 } ;
" Joint B_R: F12"
F12Sol = { F12xSol, F12ySol, 0 } ;

"∑F(2,3)=0: F03+F3+F43+F2+F12=0 (5)"
eqF23 = F03Sol+F3+F43+F2+F12Sol;
eqF23x = eqF23[[1]]==0;
eqF23y = eqF23[[2]]==0;

"∑F(2).BC=0: (F12+F2).BC=0 (6)"
rBC = ( rC - rB ) /.rule ;
eqF2BC = (F12Sol+F2).rBC == 0 ;

"∑M(3&2)_B=0: BC3xF3+BCxF03+BDxF43+M2+M3=0 (7)"
rBC3 = ( rC3 - rB ) /.rule  ;
rBC = ( rC - rB ) /.rule  ;
rBD = ( rD - rB ) /.rule  ;
eqM23B = (Cross[rBC3,F3]+Cross[rBC,F03Sol]+Cross[rBD,F43]+M2+M3)[[3]] == 0 ;


solF23 = Solve[{eqF23x,eqF23y,eqF2BC,eqM23B},
{F03xSol,F03ySol,F12xSol,F12ySol}];

F03x=F03xSol/.solF23[[1]];
F03y=F03ySol/.solF23[[1]];
F03={F03x,F03y,0};
F12x=F12xSol/.solF23[[1]];
F12y=F12ySol/.solF23[[1]];
F12={F12x,F12y,0};
Print["Eqs.(5)(6)(7) =>"];
Print["F03 = ",F03," [N]"];
Print["F12 = ",F12," [N]"];
"∑F(2)=0 => F32=-F12-F2 (4)"
F32=-F12-F2;
Print["F32 = ",F32," [N]"];


 DYNAMIC FORCE ANALYSIS VIA DYADS

rA = {0, 0, 0} [m]

rC = AC = {0, 0.06, 0} [m]

rE = AE = {0, -0.25, 0} [m]

rB = AB = {0.121244, 0.07, 0} [m]

rD = AD ={-0.149492, 0.0476701, 0} [m]

rC1 = AC1 ={0.0606218, 0.035, 0} [m]

F1 = {0.0186142, -0.0990915, 0} [N]
```

M1 = {0, 0, 0} [Nm]

rC2 = AC2 = AB = {0.121244, 0.07, 0} [m]

F2 = {0.0265917, -0.0631033, 0} [N]

M2 = {0, 0, -0.000028165} [Nm]

rC3 = AC3 = {0.0498308, 0.06411, 0} [m]

F3 = {0.0492489, -0.33315, 0} [N]

M3 = {0, 0, -0.00621962} [Nm]

rC4 = AC4 = AD ={-0.149492, 0.0476701, 0} [m]

F4 = {-0.0369367, -0.0639614, 0} [N]

M4 = {0, 0, 0.0000111583} [Nm]

rC5 = AC5 = {-0.112198, 0.223409, 0} [m]

F5 = {-0.0553516, -0.410666, 0} [N]

M5 = {0, 0, 0.00481155} [Nm]

Dyad 4 & 5

Joint E_R: F05

Joint D_R: F34

∑F(4,5)=0: F05+F5+F4+F34=0 (1)

∑F(5).DE=0: (F05+F5).DE=0 (2)

∑M(4&5)_D=0: DExF05+DC5xF5+M4+M5+M5e=0 (3)

Eqs.(1)(2)(3) =>

F05 = {268.127, 135.039, 0} [N]

F34 = {-268.035, -134.564, 0} [N]

∑F(5)=0 => F45=-F05-F5 (4)

F45 = {-268.072, -134.628, 0} [N]

Dyad 2 & 3

Joint C_R: F03

Joint B_R: F12

∑F(2,3)=0: F03+F3+F43+F2+F12=0 (5)

∑F(2).BC=0: (F12+F2).BC=0 (6)

∑M(3&2)_B=0: BC3xF3+BCxF03+BDxF43+M2+M3=0 (7)

Eqs.(5)(6)(7) =>

```
F03 = {-256.71, -272.141, 0} [N]
```

```
F12 = {-11.4013, 137.974, 0} [N]
```

```
∑F(2)=0 => F32=-F12-F2 (4)
```

```
F32 = {11.3747, -137.91, 0} [N]
```